
7 Crack Shear by Reduced Friction

7.1 Problem Statement

This verification problem involves the calculation of the total shear displacement that occurs along
a crack when the crack slips. Crack slip is induced by reducing the frictional strength of the crack.
The crack has a finite length, 2a, and is embedded in an elastic medium. The exact solution for
total shear displacement across the crack is given by

u
(tot)
s = −2(1 − ν)

G
τ (a2 − x2)1/2 (7.1)

where G = shear modulus;
ν = Poisson’s ratio;
τ = shear stress on crack surfaces;
a = half-width of crack; and
x = distance from center of crack.

Figure 7.1 illustrates the geometry conditions for the crack.

Figure 7.1 Crack slip by reduced friction

In order to compare this solution to UDEC, shear stress, τ , is calculated as a stress drop from an
initial in-situ stress caused by sliding, or

τ(drop) = μ σn − τ(initial) (7.2)

where μ is the friction coefficient of the crack, and σn is the normal stress acting on the crack.

The parameters used in this problem are

G = 109 Pa;

ν = 0.25;

μ = 0.165 (9.46◦);

τ(initial) = 107 Pa;

σn = 1.5 × 107 Pa; and

τ(drop) = 0.165(1.5 × 107) − 107

= − 7.5 × 106 Pa.

7.2 UDEC Model

An embedded crack was created in UDEC by preventing the ends of a throughgoing joint from
sliding. The crack, shown as line A-B in Figure 7.2, is represented by 22 contacts. Note that there
are actually two contacts located at each boundary gridpoint along the joint between the two blocks.
This provides a more accurate representation of edge-to-edge contacts (see Section 1.2.4 in Theory
and Background). The crack length, measured as the distance from the midpoint between the
slipping and non-slipping contacts near point A, and that near point B, is 10.32 m. The UDEC
model is composed of two blocks with a total of 768 triangular finite-difference zones. The zoning
and boundary conditions applied to the UDEC model are shown in Figure 7.2. A full boundary
element matrix is in effect around the model to simulate the infinite elastic medium.

A B

= 10

n

n

= 1.5x10 7

7

Figure 7.2 UDEC model – two blocks with triangular finite-difference zones
in each block

Shear displacements along the crack are calculated for these model conditions. The normalized
values at the crack contacts are calculated in a FISH function (crack slip), and stored versus
normalized contact location in Table 1. The values calculated from Eq. (7.1) are stored in Table 2
for comparison. See Section 7.4 for a listing of the UDEC data file for this model.

7.3 Results

The total shear displacement along the crack calculated by UDEC is plotted in Figure 7.3. The
normalized results from the UDEC simulation are plotted versus the exact solution in Figure 7.4. As
the figure indicates, the agreement is very good. Figure 7.5 displays contours of the xx-component
of the stress tensor. The stress concentration is shown near the ends of the joint.

Figure 7.3 Shear displacement along slipping crack

Figure 7.4 Comparison of results for crack shear (Table 1: UDEC, Table 2:
analytical solution)

Figure 7.5 xx-stress contours

7.4 Listing of Data File

Example 7.1 CRACK.DAT

model new
;file: crack.dat
model title ’Crack Shear by Reduced Friction’
;
block tolerance corner-round-length 0.1
block create polygon 0 0 0 12 30 12 30 0
block cut crack -1 6 31 6
block zone gen edge 1.5
block prop material 1 density 3e-3 bulk 1.667e3 shear 1.0e3
block contact prop material 1 st-n 12.6e3 st-s 5.4e3 friction 89
block contact prop material 2 st-n 12.6e3 st-s 5.4e3 friction 89
block contact change material 2
block contact change material 1 range position-x 10 20 position-y 0 12
block gridpoint history displacement-y 15 6
block gridpoint history displacement-x 15 6
model display hist 2
block edge apply stress 0.0 -1.0e1 -1.5e1
block insitu stress 0.0 -1.0e1 -1.5e1
block cycle 1
; link boundary to b.e. program
block boundary-element material 1
block boundary-element gen range position-x -1 31 position-y -1 13
block boundary-element fix 6 -1 35 0
block boundary-element stiff
; reduce friction to induce slip
block contact prop material 1 friction 9.464
block solve ratio 1e-5
;
fish define crack_slip

a_dis = 5.16
x_cen = 15.0
x_rt = 20.0
x_lt = 10.0
a_g = 1e9
a_nu = 0.25
tau_drop = -7.5e6
table(2,-1.0) = 0.0
table(2,1.0) = 0.0
ic = block.contact.head
loop while ic # 0

if block.contact.pos.x(ic) > x_lt then
if block.contact.pos.x(ic) < x_rt then

x_dis = (block.contact.pos.x(ic) - x_cen) / a_dis
u_slip = block.contact.disp.shear(ic) / a_dis
table (1,x_dis) = u_slip
a_fact = 2.0 * (1.0 - a_nu) * tau_drop / a_g
a_slip = a_fact * math.sqrt(1.0 - x_dis * x_dis)
table(2,x_dis) = a_slip

endif
endif
ic = block.contact.next(ic)

endloop
end
@crack_slip
table 1 label ’UDEC’
table 2 label ’Analyic’
;
model save ’crack.sav’
return

